


Calculating the $e^+e^- \rightarrow \mu^+\mu^-$ cross section using Schoonschip

In electron-positron colliders, one of the fundamental processes is $e^+e^- \rightarrow f^+f^-$, where f is a spin 1/2 fermion. The prototype Feynman diagram for these processes is $e^+e^- \rightarrow \mu^+\mu^-$, illustrated below



This diagram can be expressed as a complex amplitude of the form

$$\mathcal{M} = \frac{e^2 \bar{u}(\vec{p}_3) \gamma_\mu v(\vec{p}_4) \bar{v}(\vec{p}_2) \gamma_\mu u(\vec{p}_1)}{(p_1 + p_2)^2}, \quad (1)$$

where $(p_1 + p_2)^2 = -s$ and the invariant s is the square of the center of mass energy. If we imagine and e^+e^- collider with beam energies large compared with any mass, $\sqrt{s} = 2p$, where p is the beam momentum.

The Schoonschip code for the evaluation of $|\mathcal{M}|^2$ in the center of mass frame is

Date: Mon Aug 6 2001 17:59:53. Memory: start 00020008, length 476860.

```

A me,mf,cos,p,pp,Ee,Ef,s
V p1,p2,p3,p4
I mu,nu,i1,i2
F M=u
Z T=M(mu)*Conjg(M(nu))
L 2 Id,M(mu~)=Ubg(i1,mf,p3)*G(i1,mu)*Ug(i1,-mf,p4)*
      Ubg(i2,-me,p2)*G(i2,mu)*Ug(i2,me,p1)/s
L 3 Id,Spin,i1,i2
L 4 Id,Trick,Trace,i1,i2
> P out
      *yep

T = + 64*me^2*mf^2*s^-2 - 32*me^2*s^-2*p3Dp4 - 32*mf^2*s^-2*p1Dp2
      + 32*s^-2*p1Dp3*p2Dp4 + 32*s^-2*p1Dp4*p2Dp3 + 0.

C p1(mu~)=(0,0,p,iEe)
C p2(mu~)=(0,0,-p,iEe)
C p3(mu~)=(pp*sin,0,pp*cos,iEf)
C p4(mu~)=(-pp*sin,0,-pp*cos,iEf)
L 1 Id,p1Dp=-p^2-Ee^2
L 2 Id,p3Dp4=-pp^2-Ef^2
L 3 Id,p1Dp3=p*pp*cos-Ee*Ef
L 4 Id,p2Dp4=p*pp*cos-Ee*Ef

```

```

L 5      Id,p1Dp4=-p*pp*cos-Ee*Ef
L 6      Id,p2Dp3=-p*pp*cos-Ee*Ef
C Ignore masses
L 7      Id,me=0
L 7      Al,mf=0
L 8      Id,Ee=p
L 8      Al,Ef=p
L 9      Id,pp=p
          B s,p
> P out
          *end

```

and the output file is

Schoonschip, 68000 version of June 27, 1991. Public version.
Date: Mon Aug 6 2001 18:58:13. Memory: start 00020008, length 476860.

Command line: e+e-2mu+mu-.e e+e-2mu+mu-.txt

```

A me,mf,cos,p,pp,Ee,Ef,s
V p1,p2,p3,p4
I mu,nu,i1,i2
F M=u
Z T=M(mu)*Conjg(M(nu))
L 2 Id,M(mu~)=Ubg(i1,mf,p3)*G(i1,mu)*Ug(i1,-mf,p4)*
      Ubg(i2,-me,p2)*G(i2,mu)*Ug(i2,me,p1)/s
L 3 Id,Spin,i1,i2
L 4 Id,Trick,Trace,i1,i2
> P out
*yep

```

$$\begin{aligned}
T = & + 64*me^2*mf^2*s^{-2} - 32*me^2*s^{-2}*p3Dp4 - 32*mf^2*s^{-2}*p1Dp2 \\
& + 32*s^{-2}*p1Dp3*p2Dp4 + 32*s^{-2}*p1Dp4*p2Dp3 + 0.
\end{aligned}$$

```

C p1(mu~)=(0,0,p,iEe)
C p2(mu~)=(0,0,-p,iEe)
C p3(mu~)=(pp*sin,0,pp*cos,iEf)
C p4(mu~)=(-pp*sin,0,-pp*cos,iEf)
L 1 Id,p1Dp=-p^2-Ee^2
L 2 Id,p3Dp4=-pp^2-Ef^2
L 3 Id,p1Dp3=p*pp*cos-Ee*Ef
L 4 Id,p2Dp4=p*pp*cos-Ee*Ef
L 5 Id,p1Dp4=-p*pp*cos-Ee*Ef
L 6 Id,p2Dp3=-p*pp*cos-Ee*Ef
C Ignore masses
L 7 Id,me=0
L 7 Al,mf=0

```

```

L 8 Id,Ee=p
L 8 A1,Ef=p
L 9 Id,pp=p
B s,p
> P out
*end

```

```

T = + s^-2*p^4
    * ( 64 + 64*cos^2 ) + 0.

```

End run. Time 0 sec.

Including the coupling $e^4 = (4\pi\alpha)^2$, the average over initial electron spins $1/4$ and the fact that $s = 4p^2$, $|\mathcal{M}|^2$ is

$$|\mathcal{M}|^2 = 16\pi^2\alpha^2(1 + \cos^2(\theta)). \quad (2)$$

In this case, the differential phase space is

$$dPS = \frac{d\Omega}{2(4\pi)^2}, \quad (3)$$

and the flux factor is

$$4E_e^2|\vec{v}_1 - \vec{v}_2| = 4E_e^2\left(\frac{p}{E_e} + \frac{p}{E_e}\right) = 8pE_e \rightarrow 8p^2 = 2s. \quad (4)$$

Thus, $d\sigma_{\mu\bar{\mu}}/d\Omega$ is

$$\frac{d\sigma_{\mu\bar{\mu}}}{d\Omega} = \frac{\alpha^2(1 + \cos^2(\theta))}{4s}. \quad (5)$$

The total cross section is

$$\sigma_{\mu\bar{\mu}} = \frac{4\pi\alpha^2}{3s}. \quad (6)$$

In the Spear collider, the center of mass energy \sqrt{s} was 3.1 GeV, which is much larger than $m_\mu = 105.66$ MeV. This energy is also much larger than the light quarks u, d and s . The u, d and s electric charges are $2/3, -1/3$ and $-1/3$ and each quark comes in three colors. Hence, the effective coupling for light quark production is

$$3\left(\frac{2^2}{3^2} + \frac{(-1)^2}{3^2} + \frac{(-1)^2}{3^2}\right)\alpha^2 = 2\alpha^2, \quad (7)$$

Thus, we would expect to see a region where $q\bar{q}$ production was twice as large as $\mu\bar{\mu}$, or

$$R = \frac{\sigma_{q\bar{q}}}{\sigma_{\mu\bar{\mu}}} = 2. \quad (8)$$

The experimental data for R in the light quark region is shown below. At lower energies, the data shows the ρ , ω and ϕ vector mesons. Above about 1.1 GeV, there is a flat R value at $R = 2.0$. This was the first strong evidence for the quark hypothesis.




Figure 1: The R value for $\sqrt{s} \leq 3.0$ GeV is shown.